

Formatting Plots

Chris Rackauckas

March 1, 2019

Since the plotting functionality is implemented as a recipe to Plots.jl, [all of the options open to Plots.jl can be used in our plots](#). In addition, there are special features specifically for [differential equation plots](#). This tutorial will teach some of the most commonly used options. Let's first get the solution to some ODE. Here I will use one of the Lorenz ordinary differential equation. As with all commands in DifferentialEquations.jl, I got a plot of the solution by calling `solve` on the problem, and `plot` on the solution:

```
using DifferentialEquations, Plots, ParameterizedFunctions
gr()
lorenz = @ode_def Lorenz begin
    dx = σ*(y-x)
    dy = ρ*x-y-x*z
    dz = x*y-β*z
end σ β ρ
p = [10.0,8/3,28]
u0 = [1., 5., 10.]
tspan = (0., 100.)
prob = ODEProblem(lorenz, u0, tspan, p)
sol = solve(prob)


retcode: Success
Interpolation: Automatic order switching interpolation
t: 1360-element Array{Float64,1}:
 0.0
 0.0354861341350177
 0.0606639441609802
 0.10188862127423248
 0.1448494744943986
 0.19835643663680771
 0.25049906268405814
 0.3056767768178228
 0.354528003497134
 0.4077097758394896
  :
99.45257848856423
99.5206415332703
99.59156525518577
99.65749274487298
99.7343767284518
```

```

99.80001907955017
99.87708602888114
99.96443435623007
100.0
u: 1360-element Array{Array{Float64,1},1}:
[1.0, 5.0, 10.0]
[2.31565, 5.89756, 9.40679]
[3.23779, 7.04103, 9.23368]
[4.99386, 9.83293, 9.62611]
[7.42116, 13.9492, 11.5823]
[11.4597, 19.7531, 18.1042]
[15.4761, 21.5109, 29.8871]
[16.4475, 13.1242, 40.9711]
[12.8778, 2.61892, 41.2525]
[7.13698, -3.09341, 35.5052]
:
[11.856, 5.92339, 36.9333]
[7.34741, 0.974252, 32.65]
[3.75943, 0.129304, 27.1489]
[2.12612, 0.641365, 22.8278]
[1.59165, 1.53693, 18.7264]
[1.82744, 2.58043, 15.9256]
[2.79845, 4.63575, 13.5359]
[5.22745, 9.15156, 12.844]
[6.82718, 11.9177, 13.8365]

```


```
plot(sol)
```



Now let's change it to a phase plot. As discussed in the [plot functions page](#), we can use the `vars` command to choose the variables to plot. Let's plot variable `x` vs variable `y` vs variable


z :

```
plot(sol,vars=(:x,:y,:z))
```




We can also choose to plot the timeseries for a single variable:

```
plot(sol,vars=[:x])
```



Notice that we were able to use the variable names because we had defined the problem with the macro. But in general, we can use the indices. The previous plots would be:


```
plot(sol,vars=(1,2,3))  
plot(sol,vars=[1])
```



Common options are to add titles, axis, and labels. For example:


```
plot(sol,linewidth=5,title="Solution to the linear ODE with a thick line",
xaxis="Time (t)",yaxis="u(t) (in mm)",label=["X","Y","Z"])
```

Solution to the linear ODE with a thick line




Notice that series recipes apply to the solution type as well. For example, we can use a scatter plot on the timeseries:

```
scatter(sol,vars=[:x])
```




This shows that the recipe is using the interpolation to smooth the plot. It becomes abundantly clear when we turn it off using `denseplot=false`:

```
plot(sol,vars=(1,2,3),denseplot=false)
```




When this is done, only the values the timestep hits are plotted. Using the interpolation usually results in a much nicer looking plot so it's recommended, and since the interpolations have similar orders to the numerical methods, their results are trustworthy on the full interval. We can control the number of points used in the interpolation's plot using the `plotdensity` command:

```
plot(sol,vars=(1,2,3),plotdensity=100)
```



That's plotting the entire solution using 100 points spaced evenly in time.


```
plot(sol,vars=(1,2,3),plotdensity=10000)
```



That's more like it! By default it uses `100*length(sol)`, where the length is the number of internal steps it had to take. This heuristic usually does well, but unusually difficult equations it can be relaxed (since it will take small steps), and for equations with events / discontinuities raising the plot density can help resolve the discontinuity.

Lastly notice that we can compose plots. Let's show where the 100 points are using a scatter plot:


```
plot(sol,vars=(1,2,3))
scatter!(sol,vars=(1,2,3),plotdensity=100)
```



We can instead work with an explicit plot object. This form can be better for building a complex plot in a loop.

```
p = plot(sol,vars=(1,2,3))
scatter!(p,sol,vars=(1,2,3),plotdensity=100)
title!("I added a title")
```

I added a title



You can do all sorts of things. Have fun!

```
using DiffEqTutorials
DiffEqTutorials.tutorial_footer(WEAVE_ARGS[:folder],WEAVE_ARGS[:file])
```

0.1 Appendix

These benchmarks are part of the DiffEqTutorials.jl repository, found at: <https://github.com/JuliaDiffEq/DiffEqTutorials.jl>
To locally run this tutorial, do the following commands:

```
using DiffEqTutorials
DiffEqTutorials.weave_file("introduction","formatting_plots.jmd")
```

Computer Information:

```
Julia Version 1.1.0
Commit 80516ca202 (2019-01-21 21:24 UTC)
Platform Info:
  OS: Windows (x86_64-w64-mingw32)
  CPU: Intel(R) Core(TM) i7-8700 CPU @ 3.20GHz
  WORD_SIZE: 64
  LIBM: libopenlibm
  LLVM: libLLVM-6.0.1 (ORCJIT, skylake)
```

Environment:

```
JULIA_EDITOR = "C:\Users\accou\AppData\Local\atom\app-1.34.0\atom.exe" -a  
JULIA_NUM_THREADS = 6
```

Package Information:

```
Status `C:\Users\accou\.julia\external\DiffEqTutorials.jl\src\Project.toml`
```